DIMENSIONAMIENTO DE SECCIONES RECTANGULARES Y CIRCULARES DE CONCRETO SOMETIDAS A FLEXOCOMPRESION RECTA DE ACUERDO CON LAS NORMAS ACI 318-83

Por
Santiago Rizo Brenes
Catedrático
Universidad de Costa Rica
Escuela de Ingeniería Civil

Octubbre de 1988

RESUMEN

Este trabajo describe un programa de computadora para el cálculo de las columnas más comunes que se presentan en el diseño de estructuras de concreto. El programa se basa en un algoritmo general desarrollado por el autor (ref 3 y 4) denominado "Método Iterativo de las Líneas Isoaxiales" que permite calcular de manera directa y rápida el área de refuerzo de una columna de concreto, eliminando la necesidad de usar gráficas de diseño.

Se resumen las bases de la metodología propuesta, adaptándose a las especificaciones corrientes ACI 318-83. Se incluyen además varios ejemplos de aplicación de dicho programa.

RESISTENCIA DE UNA SECCION DE CONCRETO

La resistencia de una sección de concreto se obtiene a partir de las supociones básicas de la teoría de la flexocompresión que se describen en forma resumida a continuación:

1) Las secciones planas antes de la carga permanecen planas hasta la rotura, y ésta ocurre a una deformación máxima de .003 en la fibra comprimida más alejada del eje neutro.

2) Se adopta el bloque rectangular de esfuerzos para el hormigón en compresión y se desprecia su resistencia a la tracción.

3) El diagrama esfuerzo-deformación del acero es bilineal (elastoplástico), igual en tracción que en compresión, y se considera garantizada la compatibilidad de las deformaciones entre el acero de refuerzo y la masa de hormigón que lo rodea.

De acuerdo con las suposiciones anteriores y usando un solo pivote para todos los posibles diagramas de deformación, se calculan, a partir de la figura (1), la fuerza resistente P_r y el momento resistente M_r con respecto al centro geométrico:

$$P_r = 0.85f'_cA_c + \sum_{i=1}^{n} As_i fs_i \tag{1}$$

$$M_r = 0.85f'_cA_c(h/2-X_c) + \sum_{i=1}^{n} As_i fs_i (h/2-ds_i) \tag{2}$$

En estas ecuaciones f'_c es la resistencia cilíndrica del hormigón, A_c es el área de la zona comprimida con un esfuerzo $0.85f'_c$, que tiene un centroide X_c, y As_i, fs_i, ds_i son respectivamente las áreas de cada varilla, su esfuerzo y su distancia a la fibra más comprimida de la sección.

Para no complicar demasiado las expresiones anteriores no se ha indicado en forma explícita el efecto del concreto desplazado por las varillas.

ECUACIONES MODIFICADAS DE RESISTENCIA

El patrón de armado de una sección de concreto se define por el número total de varillas n y por los valores As_i,ds_i. Ya que éstos se desconocen al inicio del diseño, el primer paso consiste en fijar las distancias ds_i y suponer valores arbitrarios as_i del área de cada varilla. Estas se llaman **áreas relativas** y la suma de n áreas, as_t, **área total relativa.**

Por lo tanto, se puede escribir la siguiente relación entre áreas necesarias y relativas:

$$As_i = p\ A_b\ (as_i/as_t) \tag{3}$$

en donde p es el porcentaje de acero y A_b es el área bruta de la sección transversal.

Al sustituir As_i en las ecuaciones (1) y (2) y simplificar, se obtienen las siguientes ecuaciones modificadas de resistencia, en las cuales el porcentaje de acero aparece en forma explícita:

$$P_r = 0.85f'_cA_c + (A_b/as_t) \sum_{i=1}^{n} as_i fs_i * p \tag{4}$$

$$M_r = 0.85f'_cA_c(h/2-X_c) + (A_b/as_t) \sum_{i=1}^{n} as_i fs_i (h/2-ds_i) * p \tag{5}$$

Las secciones transversales aquí tratadas generalmente tienen varillas de un mismo diámetro. En este caso el área relativa de cada una de ellas puede suponerse igual a la unidad, y el área total relativa igual al número total de varillas n.

ECUACION DE LA RECTA ISOAXIAL:

Las ecuaciones (4) y (5) se pueden escribir en forma resumida de la siguiente manera:

$$P_r = P_c + (A_b/as_t)Ps * p \tag{6}$$

$$M_r = M_c + (A_b/as_t)Ms * p \tag{7}$$

en donde P_c, M_c es el aporte del hormigón a la resistencia total de la columna y Ps,Ms corresponden a la carga y al momento del área de acero relativa, valores que quedan determinados cuando se fija la posición del eje neutro.

En esta forma las ecuaciones (6) y (7) representan las ecuaciones paramétricas de una recta **de igual eje neutro** o **Recta Isoaxial**, siendo p el parámetro. Si éste se elimina entre ellas, la ecuación de dicha línea en el plano M,P es:

$$(P_r-P_c) = (Ps/Ms)(M_r-M_c) \tag{8}$$

en donde Ps/Ms=k es la pendiente de la recta isoaxial que pasa por el punto de coordenadas (M_c, P_c). Ver figura (2).

CALCULO DEL PORCENTAJE DE ACERO

Si en las ecuaciones paramétricas se iguala la solicitación con la resistencia y se despeja en cada una de ellas el porcentaje de acero, se tiene:

$$p = as_t(M'_u-M_c)/(A_bMs) = as_t(P'_u-P_c)/(A_bPs) \tag{9}$$

que equivale a calcular los parámetros M_c, P_c,Ms,Ps que determinan la recta isoaxial que pasa por el punto de solicitación $S(M'_u, P'_u)$, como se indica en la figura (2).

El cálculo anterior se lleva a cabo en forma iterativa, y requiere que el punto esté localizado entre dos rectas isoaxiales, una por arriba del punto, llamada superior, y otra por debajo del punto llamada inferior . Con el promedio de la posición del eje neutro de estas dos rectas, X=(Xs+Xi)/2 , se calcula una tercera recta isoaxial de tanteo. Si la distancia del punto a esta recta es pequeña el proceso se termina y se procede a calcular el porcentaje de acero. Si no, se analiza si la recta de tanteo es superior o inferior, lo cual se determina por medio el signo de la distancia. Esta recta de tanteo sustituirá a la correspondiente recta isoaxial (superior o inferior) de las dos que le dieron origen, para continuar en esta forma el proceso iterativo. Como se puede ver en la figura (3), se comienza con una recta superior ($x=h/B1$), una inferior ($x=0$) y la recta balanceada como recta de tanteo.

La expresión de la distancia se obtiene de la ecuación normal de la recta de tanteo, sustituyendo las variables por las coordenadas del punto S:

$$d = [(P'_u-P_c)-k(M'_u-M_c)]/(k^2+1)^{1/2} \qquad (10)$$

El método así planteado de denomima **Método Iterativo de Las Rectas Isoaxiales.**

EXCENTRICIDADES PEQUEÑAS

Las especificaciones requieren que la carga de solicitación de la columna sea por lo menos igual al 80% de su resistencia con carga concéntrica cuando el refuerzo transversal está constituido por aros, y de 85% cuando es en espiral. Para el caso de secciones con aros se tiene:

$$P'_u = 0.8\ [0.85f'_c(A_b-As_t) + f_yAs_t] \qquad (11)$$

Sustituyendo As_t por p^*A_b y despejando, se obtiene la expresión correspondiente para el porcentaje de acero:

$$p_m = (1.25P'_u/A_b-0.85f'_c)/(f_y-0.85f'_c) \qquad (12)$$

Cuando la columna tiene refuerzo en espiral hay que sustituir en la ecuación (11) 0.8 por 0.85 y en la ecuación (12) 1.25 por 1.18.

Si el porcenteje de acero evaluado en esta forma es menor o igual a cero, se utiliza el Método Iterativo planteado en el artículo anterior para calcular el acero definitivo.

En caso contrario se evalúa el momento M_m que acompaña a la carga de solicitacion P'_u, para el porcentaje de acero calculado con la ecuación (12).

Esta operación se efectúa en forma iterativa utilizando el Método de la Mitad. El valor obtenido de M_m se compara con el momento de solicitación M'_u; si éste es más pequeño que aquel, el porcentaje de acero correspondiente a la ecuación (12) será el definitivo; si no, se utiliza el Método de las Rectas Isoaxiales. Ver figura (4).

El procedimiento para calcular M_m puede utilizarse además para evaluar la capacidad de momento de una rótula plástica , dada una carga axial y el refuerzo longitudinal en la misma.

COEFICIENTE DE DISMINUCION DE LA CAPACIDAD

En la figura (5) se muestran los valores que toma el coeficiente de disminución de la capacidad en secciones con aros, dependiendo de las coordenadas del punto de solicitación $S(M_u,P_u)$ provenientes del analisis estructural, del esfuerzo de fluencia del acero y de la razón $(h-2r)/h$. Además interviene aquí la recta isoaxial balanceada, ya que el aumento de .7 en flexocompresión a .9 en flexión tiene sentido sólamente si la falla es dúctil.

Para secciones con espiral el valor del coeficiente de 0.7 debe sustutuirse por 0.75 y en las ecuaciones la constante de 0.2 por 0.15.

Como puede notarse en esa misma figura, excepto en la zona comprendida por el triángulo ABC, el valor del coeficiente se obtiene mediante un cálculo sencillo y directo, previo al cálculo del área de acero. No sucede lo mismo en la parte triangular, ya que allí el valor del coeficiente es función de la carga balanceada, la cual no se conoce de antemano. En este caso se adopta un procedimiento de tanteos sucesivos en el cual se parte de una carga balanceada inicial supuesta, que debe comprobarse una vez obtenida el área de acero necesaria. En el programa la carga inicial P_w se toma igual a la intersección de la recta isoaxial balanceada con la recta $M=M_u$.

Los valores de M_1, M_2 y P_w que determinan las diferentes zonas en la figura (5) son las siguentes:

$$M_1 = (0.1f'_cA_b-P_{cb})/k_b + M_{cb} \tag{13}$$

$$M_2 = -P_{cb}/k_b + M_{cb} \tag{14}$$

$$P_w = k_b(M_u-M_{cb}) + P_{cb} \tag{15}$$

que se obtienen a partir de la ecuación de la recta isoaxial balanceada, sustituyendo en ella respectivamente $P_r=0.1f'_cA_b$, $P_r=0$, $M_r=M_u$ y despejando los valores correspondientes.

Es interesante notar que cuando la solicitación se encuentra en la zona ABC, y conforme se acerca al punto C, el valor del coeficiente de disminución de la capacidad es más sensible a las variaciones de la carga de solicitación P_u, produciéndose cambios apreciables en el porcentaje de acero para pequeños incrementos de la misma. Finalmente, en el punto C se tiene una situación contradictoria, ya que el coeficiente tiene simultaneamente el valor 0.7 y 0.9.

Además, en algunos casos en que la relación $(h-2r)/h$ es pequeña, las curvas de interacción para porcentajes de acero elevados (entre el 5% y el 8%) tienden a confundirse con la recta isoaxial balanceada, lo cual produce serias discontinuidades en el cálculo del porcentaje de acero. Para coincidir con valores conservadores obtenidos utilizando la referencia (2), en el programa se optó por incrementar la fuerza de solicitación en una cierta cantidad cuando el punto queda en dicha zona,de manera que éste se ubique por encima de la recta balanceada y el coeficiente de disminución de la capacidad valga 0.7 o 0.75.

CODIFICACION DEL PROGRAMA

La codificación se hizo en BASIC por ser el lenguaje más popular utilizado en microcomputadoras y calculadoras programables. Se escogió una versión avanzada del mismo (Microsoft Corporation, 1985) con el objeto de lograr claridad de lectura, aunque no se utilizaron sus características estructuradas, para evitar incompatibilidad con otras versiones anteriores muy usadas de este mismo lenguaje.

El programa no tiene numeración de líneas y usa en su lugar etiquetas para direccionar el control del mismo. Esta característica, junto con el uso de comentarios a modo de etiqueta (líneas que comienzan con un apóstrofo) lo hace más claro y legible. Esto permite además escoger la numeración más conveniente, y facilita la localización de cálculos secuenciales sin control condicionado, que pueden escribirse en una sola línea.

USO DEL PROGRAMA

El uso del programa es sencillo y evidente y sólo requiere tener en cuenta dos cosas: 1) El recubrimiento se toma del borde de la sección al centro de la varilla longitudinal. 2) El código de armado en secciones rectangulares es un número entero entre 22 y 99, en el cual las decenas y las unidades indican respectivamente, el número de varillas en las caras perpendiculares y el número de varillas en las caras paralelas al plano de la carga.

En los ejemplos numéricos se puede notar que además del porcentaje de acero, los resultados muestran el área total de refuerzo, el número total de varillas, el diámetro de las mismas, la relación entre la carga dada y la carga balanceada, y entre la carga dada y la carga correspondiente a esfuerzo cero el las varillas más alejadas de la zona comprimida. Esto último se requiere para calcular la longitud de los empalmes de las varillas.

Finalmente se presenta un menú que permite variar uno o varios datos en forma independiente y luego efectuar los cálculos correspondientes.

COMENTARIOS

El programa equivale a las secciones 7 y 8 del Manual de Diseño de Columnas SP-17A(78), Referencia (2), que contiene 112 diagramas de interacción y 28 tablas de valores límites, y constituye un sub-programa de un programa más amplio de diseño asistido por computador (CAD), para marcos planos de concreto.

El Método Iterativo de las Rectas Isoaxiales es completamente general y puede extenderse para considerar cualquier bloque de distribución de esfuerzos del hormigón, y aplicarse a secciones poligonales cualquiera, con flexocompresion recta o esviada. Ver referencias (3),(4) y (5).

Sin embargo, cuando se requiere calcular el volumen y el centro de gravedad de una distribución compleja de esfuerzos aplicada a una zona poligonal cualquiera, es necesario contar con un algoritmo adecuado para tal propósito , como el que se indica en la referencia (6). En estos casos, la cantidad de cálculos aritméticos aumenta enormemente con la complejidad geométrica, por lo cual es recomendable usar un lenguaje compilado, como el C, el Fortran o el Pascal.

Por último, puede incluirse como una variable adicional la deformación máxima de rotura del hormigón en compresión o considerar la zona de endurecimiento del diagrama esfuerzo-deformación del acero.

BIBLIOGRAFIA

1-"Building Code Requirements for Reinforced Concrete (ACI 318-83)", American Concrete Instutute, Detroit, 1983

2-"Design Handbook (In Accordance whith the strength Design Method of ACI 318-77),Volumen 2-Columns",Publication SP-17A(78), American Concrete Institute, detroit,1978

3- Santiago Rizo. "Método Numérico para el Cálculo del Acero de Refuerzo de Columnas de Concreto",Seminario de Ingeniería Estructural,1983

4- Santiago Rizo. "Diseño de Columnas Poligonales de Concreto Sometidas a Flexocompresion Biaxial", Primer Congreso Iberoamericano de Métodos Computacionales en Ingeniería, Madrid, 1985

5- Pablo Montes de Oca. "Programa Para el Diseño de Columnas de Concreto Simétricas Con Cargas en el Plano de Simetría", Proyecto de Graduación, Escuela de Ingenieria Civil, Universidad de Costa Rica, 1984

6- Joaquín Marín. "Computing Columns, Footings and Gates Through Moments of Area". Computers & Structures, Vol. 18, No2, pag 343-349, 1984

APENDICE: LISTADO Y EJEMPLOS NUMERICOS

```
'        DISENO DE SECCIONES
'RECTANGULARES Y CIRCULARES DE CONCRETO
'   SOMETIDAS A FLEXOCOMPRESION RECTA
'        SEGUN NORMAS ACI-83
'            Octubre 1986
'        Revision:Octubre 1988
Inicio:
   CLEAR
   DIM a(10),d(10)
   pi=3.141592654#
'EscogerTipo:
   PRINT"Escoja el tipo de columna:"
   PRINT" 1 - Rectangular con Aros"
   PRINT" 2 - Cuadrada con Espiral"
   PRINT" 3 - Circular con Espiral"
   INPUT t
   IF t=1 THEN f=.7:GOTO EntradaDatos
   f=.75
EntradaDatos:
   GOSUB Esfuerzos
   GOSUB Dimensiones
   GOSUB Recubrimiento
   GOSUB Armado
   GOSUB Solicitacion
Calculos:
   pk=fr*ab:n=pk/400
   Pr=.1*fc*ab
   IF c=1 THEN GOTO RectIsoaxFsCero
   a1=ab/ar/100
   d=h-r
   ga=(h-2*r)/h
'ArregloVarillas:
   FOR I=0 TO 9:a(I)=0:NEXT
   IF t<>1 THEN ArregloCircular
   'ArregloRectangular:
      a(1)=z:a(j)=z
      d(1)=r:d(j)=d
      IF j<3 THEN InerciaRelAcero
      s=(h-2*r)/(j-1)
      FOR I=2 TO j-1
      a(I)=2:d(I)=d(I-1)+s
      NEXT I
      GOTO InerciaRelAcero
   ArregloCircular:
      FOR I=1 TO j+1
      d(I)=h/2-(h/2-r)*COS(pi*(I-1)/j)
      NEXT I
      a(1)=1:a(j+1)=1
      FOR I=2 TO j:a(I)=2:NEXT
InerciaRelAcero:
   Isr=0
   FOR I=1 TO j+1
   Isr=Isr+a(I)*(h/2-d(I))^2
   NEXT I
RectIsoaxFsCero:
   x=h-r
   GOSUB RectaIsoaxial
   g0=p0
   k0=k:Fs0=Fs
'RectIsoaxBal:
   x=6000*d/(6000+fy):xbal=x
   GOSUB RectaIsoaxial
   b0=p0*f:kb=k:Fsb=Fs
   n0=m0*f
   M1=(Pr-b0)/kb+n0
   M2=n0-b0/kb
   M3=p0+ab*Fs*8/ar
   Fw=(Um-n0)*kb+b0
CalculoFi:
   fi=.9-(.9-f)*Up/Pr:fii=0
   IF Up>=Pr THEN fi=f
   IF Up<0 THEN fi=f:GOTO AcNecesario
   q=4227.282
   IF fy<=q AND ga>=.7 THEN AcNecesario
   IF Um<=M1 THEN AcNecesario
   IF Um>=M2 THEN fi=f:GOTO AcNecesario
   Uu=Up+Pr/10
   IF Uu<Fw THEN fi=f:GOTO AcNecesario
   fi=.9-(.9-f)*Up/Pw
   fii=fi
AcNecesario:
   Mu=Um/fi:Pu=Up/fi:ex=ABS(Um/Up):em=h/10
   pp=pk+ab*8/100*fy
   IF Pu>pp THEN PRINT"%>8":GOTO Modificar
   'ExcentrMin:
      EnTraccion:
      cond=Pu<0 AND ex<em
      IF cond THEN Mu=-Pu*em:GOTO Flexocompr
      EnCompresion:
      ro=100*(Pu/(f+.1)/ab-fr)/(fy-fr)
      IF ro<=0 THEN Flexocompr
      GOSUB CalcMuDadoPu
      IF Mu<mm THEN CargaBal
   Flexocompr:
      xs=h/b1:xi=0
      flag=0
      M.Mit:
         x=(xs+xi)/2
         IF flag=0 THEN x=xbal:flag=1
         GOSUB RectaIsoaxial
         g=Pu-p0-k*(Mu-m0)
         IF Ms<0 THEN g=-g
         dd=ABS(g/SQR(k^2+1))
         IF dd<=n THEN CalculoAc
         IF SGN(g)>0 THEN xi=x:GOTO M.Mit
         xs=x:GOTO M.Mit
      CalculoAc:
      rp=(Pu-p0)/a1/Fs
      IF x<r THEN ro=rp:GOTO CargaBal
      ro=(Mu-m0)/a1/Ms
 CargaBal:
    IF ro<=0 THEN Imprimir
    pb=b0/f+a1*Fsb*ro
    bp=Pu/pb
    IF fii=0 THEN CalcAdicionales
    fi=.9-(.9-f)*Up/(pb*fi)
    Dfi=ABS(fii-fi)
    IF Dfi>.01 THEN fii=fi:GOTO AcNecesario
```

```
 cAdicionales:
   Ps=g0+a1*Ps0*ro:bs=Pu/Ps
   at=ab*ro/100
   dv=20*SQR(at/ar/pi)
   Is=Isr*at/ar
   EIa=Ec*Ig/5+Es*Is
   EIb=Ec*Ig/2.5
   IF EIa>EIb THEN ei=EIa:GOTO Imprimir
   ei=EIb
 rimir:
   PRINT"RESULTADOS:"
   PRINT USING "%ac=#.##";ro
   IF ro<=0 THEN Modificar
   PRINT USING "Ast(cm2)=###.##";at
   PRINT"# de Varillas=";ar
   PRINT USING "Diam Var (mm)=###.#";dv
   PRINT USING "Pu/Pb=##.#";bp
   PRINT USING "Pu/Ps0=##.#";bs
   PRINT USING "fi=#.##";fi
   PRINT "EI (cm4)=";ei
Modificar:
   PRINT"Modificar:"
   PRINT"(E)sf,(D)im,(R)ec,(I)nic"
   PRINT"(A)rm,(S)ol,(C)al,(T)er"
 oger:
   INPUT m$:IF m$=" " THEN Escoger
   IF er=1 THEN er=0:GOTO Casos
   c=1
   Casos:
   IF m$="e" THEN GOSUB Esfuerzos
   IF m$="d" THEN c=0:GOSUB Dimensiones
   IF m$="r" THEN c=0:GOSUB Recubrimiento
   IF m$="a" THEN c=0:GOSUB Armado
   IF m$="s" THEN GOSUB Solicitacion
   IF m$="c" THEN GOTO Calculos
   IF m$="i" THEN GOTO Inicio
   IF m$="t" THEN END
   er=1:GOTO Escoger

       ****** Subrutinas ******
 uerzos:
   INPUT"f'c,fy (kg/cm2)";fc,fy
   b1=1.05-fc/1400
   IF fc<=280 THEN b1=.85
   IF b1<.65 THEN b1=.65
   Es=2043200!:Ec=15000*SQR(fc)
   ey=f /Es:fr=.85*fc
   RET  
Dimensi nes:
   IF t 3 THEN Circ
  'Rect:
      IF t=1 THEN INPUT"b,h (cm)";b,h
      IF t=2 THEN INPUT"h (cm)";h:b=h
      ab=b*h:Ig=b*h^3/12
      RETURN
   Circ:
      INPUT"d (cm)";h
      ab=pi*h*h/4:Ig=pi*h^4/64
      RETURN
Recubrimiento:
   INPUT"Recub imiento (cm)";r
   RETURN
Armado:
   IF t<>1 THEN ArmCirc
   ArmRec:
      INPUT"Armado";j:e=j
      IF j-INT   <>0 OR j>99 THEN ArmRec
      z=INT(j/10)
      j=j-10*z
      IF z<2 OR j<2 THEN ArmRec
      ar=(j+z)*2-4
      RETURN
   ArmCirc:
       PUT"# Varillas";ar
      IF ar>16 THEN ArmCirc
      j=ar/2
      IF j-INT(j)<>0 OR j<3 THEN ArmCirc
      RETURN
Solicitacion:
   INPUT"Mu,Pu (t-m,t)";Um,Up
   IF Um<0 THEN Solicitacion
   Um=Um*100000!:Up=Up*1000
   IF Up=0 THEN Up=.001
   IF Um=0 THEN Um=.001
   RETURN
RectaIsoaxial:
  'Punto0:
     a=b1*x
     IF t=3 THEN SegmentoCircular
    'AreaRectang:
       p0=fr*b*a:m0=p0*(h-a)/2
       GOTO Pendiente
     SegmentoCircular:
       co=1-2*a/h
       al=-ATN(co/SQR(1-co*co))+1.57
       p0=fr*h*h*(al-SIN(al)*COS(al))/4
       m0=fr*(h*SIN(al))^3/12
   Pendiente:
     Ps=0:Ms=0
     FOR I=1 TO j+1
       est=.003*(1-d(I)/x)
       fs=fy*SGN(est)
       IF ABS(est)<ey THEN fs=est*Es
       IF d(I)<=a THEN fs=fs-fr
       P=a(I)*fs
       Ps=Ps+P
       Ms=Ms+P*(h/2-d(I))
     NEXT I
     IF Ms=0 THEN Ms=.001
     IF Ps=0 THEN Ps=.001
     I=Ps/Ms
   RETURN
CalcMuDadoPu:
   xs=h/b1:xi=0
   metmit:
     x=(xs+xi)/2
     GOSUB RectaIsoaxial
     pc=p0+ro*Ps*a1
     IF ABS(Pu-pc)<n THEN mm
     IF pc<Pu THEN xi=x:GOTO metmit
     xs=x:GOTO metmit
     mm:mm=m0+ro*Ms*a1
   RETURN
```

EJEMPLO 3:

```
Escoja el tipo de seccion:
 1 - Rectangular con Aros
 2 - Cuadrada con Espiral
 3 - Circular con Espiral
? 2
ENTRADA DE DATOS:
f'c,fy (Kg/cm2) ? 210, 2300
h (cm) ? 30
Recubrimiento (cm) ? 6.72
# Varillas ? 6
Mu,Pu (t-m,t) ? 4.1, 10.4
RESULTADOS:
%ac=1.93
Ast(cm2)= 17.36
# Varillas= 6
Diam Var (mm)= 19.2
Pu/Pb= 0.2
Pu/Ps0= 0.1
fi=0.75
EI (cm4)= 5.869008E+09
Modificar:
(E)sf,(D)im,(R)ec
(A)rm,(S)ol,(C)alc
?
```



EJEMPLO 4:

```
Escoja el tipo de seccion:
 1 - Rectangular con Aros
 2 - Cuadrada con Espiral
 3 - Circular con Espiral
? 3
ENTRADA DE DATOS:
f'c,fy (Kg/cm2) ? 280, 2800
d (cm) ? 60
Recubrimiento (cm) ? 8
# Varillas ? 6
Mu,Pu (t-m,t) ? 26, .000001
RESULTADOS:
%ac=1.62
Ast(cm2)= 45.72
# Varillas= 6
Diam Var (mm)= 31.1
Pu/Pb= 0.0
Pu/Ps0= 0.0
fi=0.90
EI (cm4)= 6.3871121E+10
Modificar:
(E)sf,(D)im,(R)ec
(A)rm,(S)ol,(C)alc
?
```





FIGURA 1



FIGURA 2



FIGURA 4



FIGURA 3



FIGURA 5